/* Import Data – Library: statcrm */
proc import out = statcrm.customer_acquisition


datafile = "F:\_Your_Data_Location_\Statistics in CRM\Customer Acquisition\Customer Acquisition.xls" dbms = excel replace;

    range = "'Customer Acquisition Data$'"; getnames = yes; mixed = no; scantext = yes; usedate = yes; scantime = yes;

run; quit;

/* Response Probability */
proc logistic data = statcrm.customer_acquisition descending;

model acquisition = acq_expense acq_expense_sq industry revenue employees / rsquare;

output out = statcrm.response_probability_pred p = pred;

run; quit;

/* Number of Acquired Customers */
data statcrm.response_probability_pred1;

set statcrm.response_probability_pred;

if pred >= 0.5 then pred_acq = 1; else pred_acq = 0;

act_acq = acquisition;

run; quit;

proc freq data = statcrm.response_probability_pred1;

table pred_acq * act_acq; run; quit;

/* Initial Order Quantity */
proc logistic data = statcrm.customer_acquisition descending;

model acquisition =  acq_expense acq_expense_sq industry revenue employees / link = probit;

output out = statcrm.first_purchase_probit xbeta = xb_probit;

run; quit;

data statcrm.first_purchase_imr;

set statcrm.first_purchase_probit;

imr_acquisition = (pdf('Normal',xb_probit))/(probnorm(xb_probit));

run; quit;

proc reg data = statcrm.first_purchase_imr;

model first_purchase = acq_expense acq_expense_sq industry revenue employees imr_acquisition;

output out = statcrm.first_purchase_imr1 p = xbeta; 
where first_purchase > 0;

run; quit;

data statcrm.first_purchase1;

set statcrm.first_purchase_imr1;

pred_fp = probnorm(xb_probit)*(xbeta);

ad = abs(first_purchase - pred_fp);

ape = ad/first_purchase;

ad1 = abs(first_purchase - 372.47);

ape1 = ad1/first_purchase;

run; quit;

proc sql; select mean(ad) as mad, mean(ape) as mape, 

mean(ad1) as random_mad, mean(ape1) as mape1 

from statcrm.first_purchase1; quit;

/* Duration - Time */
proc lifereg data = statcrm.customer_acquisition;

model duration*censor(1) =  acq_expense acq_expense_sq ret_expense ret_expense_sq crossbuy frequency frequency_sq industry revenue employees;

where acquisition = 1; 

output out = statcrm.duration xbeta = xb p = pred sres = resid;

run; quit;

data statcrm.duration1;

set statcrm.duration;

pred_duration = exp(xb+0.138*(log(-log(1-0.5))));

ad = abs(duration - pred_duration);


ad1 = abs(duration - 333.3165);

run; quit;

proc sql; select mean(duration) from statcrm.duration1 where acquisition = 1 and censor = 0; quit;

proc sql; select mean(ad) as mad, (mean(ad/duration)) as mape, 

mean(ad1) as random_mad, (mean(ad1/duration)) as mape1 

from statcrm.duration1 where acquisition = 1 and censor = 0; quit;

/* Firm Performance */
proc logistic data = statcrm.customer_acquisition descending;

model censor = 
acq_expense acq_expense_sq ret_expense ret_expense_sq first_purchase crossbuy frequency frequency_sq industry revenue employees / link = probit;

output out = statcrm.clv1a xbeta = xb_censor;

where acquisition = 1;

run; quit;

data statcrm.clv1b;

set statcrm.clv1a;

imr_censor = pdf('Normal',xb_censor)/probnorm(xb_censor);

run; quit;

proc reg data = statcrm.clv1b;

model clv = acq_expense acq_expense_sq ret_expense ret_expense_sq first_purchase 


crossbuy frequency frequency_sq industry revenue employees imr_censor;

where clv > 0;

output out = statcrm.clv1c p = pred_clv;

run; quit; 

data statcrm.clv1d;

set statcrm.clv1c;

ad = abs(pred_clv - clv);


ad1 = abs(pred_clv - 6.58);

run; quit;

proc sql; select mean(clv) from statcrm.clv1d where censor = 1; quit;

proc sql; select mean(ad) as mad, (mean(ad/clv)) as mape, 

mean(ad1) as random_clv, (mean(ad1/clv)) as mape1 

from statcrm.clv1d where acquisition = 1 and duration = 730; quit;

